
Mappings
Images can be applied to an object, not to create a basic texture on that object, but
to further affect the final appearance of that object.

The SET LIGHT MAPPING ON Statement

Not all parts of an object's surface reflect equal amounts of light, and although
DarkBASIC Pro will take care of this for us automatically, in many cases there are
occasions when we need to give it a helping hand. We can modify how parts of an
object reflect light using a light map. A light map is a grey-scale image which is
applied to an object in much the same way as a texture image. Once wrapped onto
the object, the light map affects how light is reflected from that object. Areas of the
surface which correspond to white parts in the image reflect light as normal, while
black areas reflect little or no light. The images in FIG-39.12 show the results of a
street at night with and without light mapping.

FIG-39.12

Using Light Mapping

No light mapping applied

Light mapping applied

The default directional light is switched off and ambient light remains on in both
scenes. In the second scene, light maps have been applied to the wall, street and
lamppost.

The function, CreateScene(), which creates the first version of the scene as shown
in FIG-39.12 is given below:

FUNCTION CreateScene()
REM *** Create cobbled street ***
MAKE OBJECT PLAIN 1,30,30
XROTATE OBJECT 1,90
LOAD IMAGE "cobbleslarge.jpg",1
TEXTURE OBJECT 1,1
POSITION OBJECT 1,0,-10,0
REM *** Create wall ***
MAKE OBJECT PLAIN 2,30,20
LOAD IMAGE "brickslarge.jpg",2
TEXTURE OBJECT 2,2
POSITION OBJECT 2,0,-1,5
REM *** Create lamppost ***
MAKE OBJECT CYLINDER 3,10
SCALE OBJECT 3, 5,100,5
POSITION OBJECT 3,0,-5,0
LOAD IMAGE "lamppost.jpg",3
TEXTURE OBJECT 3,3
MAKE OBJECT SPHERE 4,2,100,100
REM *** Make sphere appear to glow ***
SET OBJECT EMISSIVE 4,RGB(255,255,255)
REM *** Put light in sphere ***
MAKE LIGHT 1
SET POINT LIGHT 1,0,0,0

ENDFUNCTION

Activity 39.10

Write a program to execute the function given above (lightmap.dbpro).

The main section should include the following logic:

Set the backdrop colour to RGB(20,20,20) - a dull grey.
Position the camera at (20,-4,-5) pointing at (0,0,3)

Test the program and observe the results.

The light map images used to create the second version of the scene are shown in
FIG-39.13.

FIG-39.13

Examples of Light Maps

Light map for wall and street
( )lightmap.jpg

Light map for lamppost
( )lampmap.jpg

We can apply light mapping to a specific object using the SET LIGHT MAPPING
ON statement which has the format shown in FIG-39.14.

In the diagram:

objno is an integer value specifying the ID of the
object to which the light map is to be applied.

imgno is an integer value giving the ID of the image
object containing the light map picture.

The program in LISTING-39.4 uses the scene produced by CreateScene() and the
light maps shown in FIG-39.13 to create the effect shown in the second image of
FIG-39.12.

REM *** Set up screen ***
SET DISPLAY MODE 1280,1024,32
COLOR BACKDROP RGB(20,20,20)
BACKDROP ON

REM *** Position camera ***
AUTOCAM OFF
POSITION CAMERA 20,-4,-5
POINT CAMERA 0,0,0

REM *** Make scene ***
CreateScene()
REM *** Load and apply light map to street and wall ***
WAIT KEY
LOAD IMAGE "lightmap.jpg",4
SET LIGHT MAPPING ON 1,4
SET LIGHT MAPPING ON 2,4

REM *** Load light map for lamppost ***
WAIT KEY
LOAD IMAGE "lampmap.jpg",5
SET LIGHT MAPPING ON 3,5

REM *** End program ***
WAIT KEY
END

REM ************ Code for CreateScene goes here ***

Activity 39.11

Type in and test the program given above (advanced04.dbpro). Remember to
include the actual code for function CreateScene().

Load and examine the file lightgrey.jpg.

Modify your program to use this file as the light map for the wall and ground.

How does this affect the result?

Unfortunately, if a 3D object is textured using tiling, the light map will also be tiled.

LISTING-39.4

Using a Light Map

FIG-39.14

The SET LIGHT
MAPPING ON Statement

MAPPINGLIGHT MAPPINGLIGHT objnoobjnoSET ONSET ON ,, imgnoimgno

